Mecánica de Medios Continuos. Ingeniero Geólogo (curso 3º). 2006-07

LABORATORIO DE MECÁNICA DE MEDIOS CONTINUOS

Tercera Sesión. Elasticidad

Miércoles 7 de Marzo de 2007 (8:30 - 10:30)

Javier Rodríguez

Nombre del alumno:
Documento Nacional de Identidad:
Por favor, guarde este fichero con el nombre <DNI>.mws, siendo <DNI> el número de su Documento Nacional de Identidad.

Resuelva las cuestiones indicadas con asterisco (*)

Problema 3.1 (2005/06)

En un punto de una zapata de hormigón de una cimentación se tiene un estado de deformación plana, midiéndose las deformaciones siguientes: epsilon_xx = 10^(-3), epsilon_yy = -2*10^(-3), epsilon_xy = 10^(-3). El material puede considerarse elástico lineal, con módulo de Young 30 GPa y de Poisson nu = 0,3. Se pide:

1) Expresar las componentes del tensor de tensiones.

```
> restart;
> with(LinearAlgebra):
> epsilon := Matrix([[1e-3, 1e-3, 0],[1e-3, -2e-3, 0],[0, 0, 0]],shape=symmetric);
```

$$\varepsilon := \begin{bmatrix} 0.001 & 0.001 & 0 \\ 0.001 & -0.002 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

```
> epsilon_v := Trace(epsilon);
```

$$epsilon\_v := -0.001$$

```
> sigma := 2*mu*epsilon + lambda*epsilon_v*IdentityMatrix(3);
```

$$\sigma := \begin{bmatrix} 0.002\,\mu - 0.001\,\lambda & 0.002\,\mu & 0 \\ 0.002\,\mu & -0.004\,\mu - 0.001\,\lambda & 0 \\ 0 & 0 & -0.001\,\lambda \end{bmatrix}$$

```
> lambda:=E*nu/((1+nu)*(1-2*nu));
```

$$\lambda := \frac{E\,\nu}{(1+\nu)\,(1-2\,\nu)}$$

```
> mu:=E/(2*(1+nu));
```

$$\mu := \frac{E}{2+2\,\nu}$$

```
> E := 30e9; nu := 0.3;
```

$$E := 0.30\ 10^{11}$$

$$\nu := 0.3$$

```
> sigma;
```

$$\begin{bmatrix} 0.576923077\ 10^{7} & 0.2307692308\ 10^{8} & 0 \\ 0.2307692308\ 10^{8} & -0.6346153847\ 10^{8} & 0 \\ 0 & 0 & -0.1730769231\ 10^{8} \end{bmatrix}$$

2) Tensi´on media, tensiones desviadoras e invariante J2

```
> sigma_m := 1/3 * Trace(sigma);
```

$$sigma\_m := -0.2500000000\ 10^{8}$$

```
> sigma_d := sigma - sigma_m * IdentityMatrix(3);
```

$$sigma\_d := \begin{bmatrix} 0.3076923077\ 10^{8} & 0.2307692308\ 10^{8} & 0. \\ 0.2307692308\ 10^{8} & -0.3846153847\ 10^{8} & 0. \\ 0. & 0. & 0.769230769\ 10^{7} \end{bmatrix}$$

```
> J[2] = 1/2 * Trace(sigma_d . sigma_d);
```

$$J_2 = 0.1775147930\ 10^{16}$$

3) Máximas y mínimas tensiones normales y direcciones respectivas.

Aplicamos la propiedad de que los autovectores de epsilon y sigma son los mismos

```
> ev := Eigenvectors(epsilon);
```

$$ev := \begin{bmatrix} -0.00230277563773199478 \\ 0. \\ 0.00130277563773199454 \end{bmatrix},$$

$$\begin{bmatrix} -0.289784148688430054 & 0. & -0.957092026489052894 \\ 0.957092026489052894 & 0. & -0.289784148688430054 \\ 0. & 1. & 0. \end{bmatrix}$$

Extraemos los autovectores y sus autovalores correspondientes (deformaciones principales)

```
> ev1 := Column(ev[2],1); lambda1 := ev[1][1];
  ev2 := Column(ev[2],2); lambda2 := ev[1][2];
  ev3 := Column(ev[2],3); lambda3 := ev[1][3];
```

$$ev1 := \begin{bmatrix} -0.289784148688430054 \\ 0.957092026489052894 \\ 0. \end{bmatrix}$$

$$\lambda 1 := -0.00230277563773199478$$

$$ev2 := \begin{bmatrix} 0. \\ 0. \\ 1. \end{bmatrix}$$

$$\lambda 2 := 0.$$

$$ev3 := \begin{bmatrix} -0.957092026489052894 \\ -0.289784148688430054 \\ 0. \end{bmatrix}$$

$$\lambda 3 := 0.00130277563773199454$$

Obtenemos la tensi´on principal 1, aplicandola a la direcci´on principal 1:

```
> sev1 := sigma . ev1;
  Lambda1 := sev1[1]/ev1[1];
```

$$sev1 := \begin{bmatrix} 0.001334615756\,\mu + 0.0002897841487\,\lambda \\ -0.004407936403\,\mu - 0.0009570920265\,\lambda \\ 0. \end{bmatrix}$$

$$\Lambda 1 := -0.7044866857\; 10^{8}$$

Comprobaci´on de que efectivamente es principal de las tensiones:

```
> sigma . ev1 = Lambda1*ev1;
```

$$\begin{bmatrix} 0.001334615756\,\mu + 0.0002897841487\,\lambda \\ -0.004407936403\,\mu - 0.0009570920265\,\lambda \\ 0. \end{bmatrix} = \begin{bmatrix} 0.204149074477908053\; 10^{8} \\ -0.674258589651169479\; 10^{8} \\ 0. \end{bmatrix}$$

Calcular los otros dos autovalores de sigma de la misma manera

```
> sva := Eigenvalues(sigma);
```

$$sva := \begin{bmatrix} -0.7044866857\; 10^{8} \\ -0.1730769231\; 10^{8} \\ 0.1275636087\; 10^{8} \end{bmatrix}$$

Problema 3.2 (2004/05)

Considérese una barra elástica de sección cuadrada de lado a y longitud L. Las constantes elásticas son E y nu = 1/4. Se pide

a) Para el caso de tracción pura de axil P, calcular la energía elástica de deformación. Determinar la componente debida al cambio de volumen y la correspondiente al cambio de forma.

```
> restart;
> sigma_xx := P/A;
```

$$sigma\_xx := \frac{P}{A}$$

```
> A := a^2;
```

$$A := a^2$$

```
> epsilon_xx := sigma_xx/E;
```

$$epsilon\_xx := \frac{P}{a^2\,E}$$

```
> Uo := 1/2*sigma_xx*epsilon_xx;
```

$$Uo := \frac{P^2}{2\,a^4\,E}$$

```
> sigma_m := 1/3*sigma_xx;
```

$$sigma\_m := \frac{P}{3\,a^2}$$

```
> epsilon_m := sigma_m/K;
```

$$epsilon\_m := \frac{P}{3\,a^2\,K}$$

```
> K := E/(3*(1-2*nu));
```

$$K := \frac{E}{3-6\,\nu}$$

```
> nu := 1/4;
```

$$\nu := \frac{1}{4}$$

```
> Uoe := 1/2*sigma_m*epsilon_m;
```

$$Uoe := \frac{P^2}{12\,a^4\,E}$$

```
> Uod := Uo - Uoe;
```

$$Uod := \frac{5\,P^2}{12\,a^4\,E}$$

```
> U := Uo*A*L;
```

$$U := \frac{P^2\,L}{2\,a^2\,E}$$

```
> Ue := Uoe*A*L;
```

$$Ue := \frac{P^2\,L}{12\,a^2\,E}$$

```
> Ud := Uod*A*L;
```

$$Ud := \frac{5\,P^2\,L}{12\,a^2\,E}$$

b) Idem para el caso de flexión pura de momento M

```
> sigma_xx := M*y/Inercia;
```

$$sigma\_xx := \frac{M\,y}{Inercia}$$

```
> Inercia := 1/12*a^4;
```

$$Inercia := \frac{a^4}{12}$$

```
> epsilon_xx := sigma_xx/E;
```

$$epsilon\_xx := \frac{12\,M\,y}{a^4\,E}$$

```
> U := a*L*int(1/2*sigma_xx*epsilon_xx, y=-a/2..a/2);
```

$$U := \frac{6\,L\,M^2}{a^4\,E}$$

```
> sigma_m := sigma_xx/3;
```

$$sigma\_m := \frac{4\,M\,y}{a^4}$$

```
> epsilon_m := sigma_m/K;
```

$$epsilon\_m := \frac{6\,M\,y}{a^4\,E}$$

```
> Ue := a*L*int(1/2*sigma_m*epsilon_m, y=-a/2..a/2);
```

$$Ue := \frac{L\,M^2}{a^4\,E}$$

```
> Ud := U-Ue;
```

$$Ud := \frac{5\,L\,M^2}{a^4\,E}$$

Problema 1*

En un punto de un sólido se conocen las componentes del tensor de deformaciones lineal epsilon = [[2,3,0],[3,-6,0],[0,0,0]]*1e-3.
El sólido es elástico lineal e isótropo con módulos elásticos E=20MPa, nu=1/4, Se pide:

a) Obtener las deformaciones principales y sus direcciones

```
> restart;
> with(LinearAlgebra):
> epsilon := Matrix([[2,3,0],[3,-6,0],[0,0,0]]*1e-3);
```

$$\varepsilon := \begin{bmatrix} 0.002 & 0.003 & 0. \\ 0.003 & -0.006 & 0. \\ 0. & 0. & 0. \end{bmatrix}$$

```
> Eigenvectors(epsilon);
```

$$\begin{bmatrix} 0.00300000000000000006 + 0.\,I \\ -0.00700000000000000014 + 0.\,I \\ 0. + 0.\,I \end{bmatrix},$$

$$\begin{bmatrix} 0.948683298050513768 + 0.\,I & -0.316227766016837885 + 0.\,I & 0. + 0.\,I \\ 0.316227766016837885 + 0.\,I & 0.948683298050513768 + 0.\,I & 0. + 0.\,I \\ 0. + 0.\,I & 0. + 0.\,I & 1. + 0.\,I \end{bmatrix}$$

b) Calcular las componentes del tensor de tensiones

```
> sigma :=
  lambda*IdentityMatrix(3)*Trace(epsilon)+2*mu*epsilon;
```

$$\sigma := \begin{bmatrix} -0.004\,\lambda + 0.004\,\mu & 0.006\,\mu & 0. \\ 0.006\,\mu & -0.004\,\lambda - 0.012\,\mu & 0. \\ 0. & 0. & -0.004\,\lambda \end{bmatrix}$$

```
> E := 20E6;
  nu := 1/4;
```

$$E := 0.20\;10^8$$

$$\nu := \frac{1}{4}$$

```
> lambda := E*nu/((1+nu)*(1-2*nu));
  mu := E/(2*(1+nu));
```

$$\lambda := 0.8000000000\ 10^{7}$$

$$\mu := 0.8000000000\ 10^{7}$$

```
> sigma;
```

$$\begin{bmatrix} 0. & 48000.00000 & 0. \\ 48000.00000 & -128000.0000 & 0. \\ 0. & 0. & -32000.00000 \end{bmatrix}$$

c) Calcular las tensiones principales y sus direcciones

```
> Eigenvectors(sigma);
```

$$\begin{bmatrix} 16000. + 0.\,I \\ -144000. + 0.\,I \\ -32000. + 0.\,I \end{bmatrix},$$

$$\begin{bmatrix} 0.948683297999999952 + 0.\,I & -0.316227765900000013 + 0.\,I & 0. + 0.\,I \\ 0.316227766000000021 + 0.\,I & 0.948683297999999952 + 0.\,I & 0. + 0.\,I \\ 0. + 0.\,I & 0. + 0.\,I & 1. + 0.\,I \end{bmatrix}$$

d) Calcular la densidad de energía elástica volumétrica, desviadora y total

```
> sigma_m := Trace(sigma)/3;
```

$$sigma\_m := -53333.33333$$

```
> epsilon_v := Trace(epsilon);
```

$$epsilon\_v := -0.004$$

```
> Wv := 1/2*sigma_m*epsilon_v;
```

$$Wv := 106.6666666$$

```
> Wt := 0;
  for i from 1 to 3 do;
    for j from 1 to 3 do;
      Wt := Wt + 1/2*sigma[i,j]*epsilon[i,j];
    od;
  od;
  Wt;
```

$$Wt := 0$$

$$528.0000000$$

```
> Wd := Wt-Wv;
```

$$Wd := 421.3333334$$

Comprobación de Wd

```
> sigma_d := sigma - sigma_m*IdentityMatrix(3);
```

$$sigma\_d := \begin{bmatrix} 53333.33333 & 48000.00000 & 0. \\ 48000.00000 & -74666.66667 & 0. \\ 0. & 0. & 21333.33333 \end{bmatrix}$$

```
> epsilon_d := epsilon - epsilon_v/3*IdentityMatrix(3);
```

$$epsilon\_d := \begin{bmatrix} 0.0033333333319999966, & 0.0030000000000000006, & 0. \\ 0.0030000000000000006, & -0.0046666666680000050, & 0. \\ 0., & 0., & 0.0013333333319999984 \end{bmatrix}$$

```
> Wd := 0;
  for i from 1 to 3 do;
    for j from 1 to 3 do;
      Wd := Wd + 1/2*sigma_d[i,j]*epsilon_d[i,j];
    od;
  od;
  Wd;
```

$$Wd := 0$$

$$421.3333332$$

Obsérvese que

```
> sigma_d :=
  lambda*IdentityMatrix(3)*Trace(epsilon_d)+2*mu*epsilon_d;
```

$$sigma\_d := \begin{bmatrix} 53333.3333231999932 & 48000. & 0. \\ 48000. & -74666.6666768000141 & 0. \\ 0. & 0. & 21333.3333231999968 \end{bmatrix}$$

Problema 4

Se considera un material elástico lineal isótropo sometido a un estado de deformación plana, con compresión vertical a y cortante a.sqrt(3)/2.
Los módulos elásticos son E=1E3, nu=1/3. Se pide:

a) Expresar las componentes del tensor de tensiones y calcular las tensiones principales y sus direcciones.
Calcular las deformaciones principales y sus direcciones.

```
> restart;
> with(LinearAlgebra):
> sigma :=
  Matrix([[0,a*sqrt(3)/2,0],[a*sqrt(3)/2,-a,0],[0,0,sigma_zz
  ]]);
```

$$\sigma := \begin{bmatrix} 0 & \frac{a\sqrt{3}}{2} & 0 \\ \frac{a\sqrt{3}}{2} & -a & 0 \\ 0 & 0 & sigma\_zz \end{bmatrix}$$

```
> epsilon :=
  Matrix([[epsilon_xx,epsilon_xy,0],[epsilon_xy,epsilon_yy,0
  ],[0,0,0]]);
```

$$\varepsilon := \begin{bmatrix} epsilon\_xx & epsilon\_xy & 0 \\ epsilon\_xy & epsilon\_yy & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

```
> sigma_lame :=
```

```
lambda*IdentityMatrix(3)*Trace(epsilon)+2*mu*epsilon;
```

$$sigma\_lame := \begin{bmatrix} \lambda\,(epsilon\_xx + epsilon\_yy) + 2\,\mu\,epsilon\_xx\,, & 2\,\mu\,epsilon\_xy\,, & 0 \\ 2\,\mu\,epsilon\_xy\,, & \lambda\,(epsilon\_xx + epsilon\_yy) + 2\,\mu\,epsilon\_yy\,, & 0 \\ 0\,, & 0\,, & \lambda\,(epsilon\_xx + epsilon\_yy) \end{bmatrix}$$

```
> lambda := E*nu/((1+nu)*(1-2*nu));
  mu := E/(2*(1+nu));
```

$$\lambda := \frac{E\,\nu}{(1+\nu)\,(1-2\,\nu)}$$

$$\mu := \frac{E}{2+2\,\nu}$$

```
> E := 1E3*a;
  nu := 1/3;
```

$$E := 1000.\,a$$

$$\nu := \frac{1}{3}$$

```
> sols := solve({sigma[1,1]=sigma_lame[1,1],
         sigma[1,2]=sigma_lame[1,2],
         sigma[2,2]=sigma_lame[2,2],
         sigma[3,3]=sigma_lame[3,3]},
     {epsilon_xx,epsilon_xy,epsilon_yy,sigma_zz});
```

$$sols := \{epsilon\_xx = 0.0004444444444,\ epsilon\_yy = -0.0008888888889,$$
$$sigma\_zz = -0.3333333333\,a,\ epsilon\_xy = 0.001154700538\}$$

```
> sigma := subs(sols,sigma);
```

$$\sigma := \begin{bmatrix} 0 & \frac{a\sqrt{3}}{2} & 0 \\ \frac{a\sqrt{3}}{2} & -a & 0 \\ 0 & 0 & -0.3333333333\,a \end{bmatrix}$$

```
> epsilon := subs(sols,epsilon);
```

$$\varepsilon := \begin{bmatrix} 0.0004444444444 & 0.001154700538 & 0 \\ 0.001154700538 & -0.0008888888889 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

```
> ev := Eigenvectors(epsilon);
```

$$ev := \begin{bmatrix} 0.001111111111074655854 + 0.\,I \\ -0.001555555555524655850 + 0.\,I \\ 0. + 0.\,I \end{bmatrix},$$
$$\begin{bmatrix} 0.866025403817287210 + 0.\,I & -0.499999999943104678 + 0.\,I & 0. + 0.\,I \\ 0.499999999943104678 + 0.\,I & 0.866025403817287210 + 0.\,I & 0. + 0.\,I \\ 0. + 0.\,I & 0. + 0.\,I & 1. + 0.\,I \end{bmatrix}$$

```
> sigma_1 :=
  simplify(Transpose(Column(ev[2],1)).sigma.Column(ev[2],1))
```

```
;
sigma_2 :=
simplify(Transpose(Column(ev[2],2)).sigma.Column(ev[2],2))
;
sigma_3 :=
simplify(Transpose(Column(ev[2],3)).sigma.Column(ev[2],3))
;
```

$$sigma\_1 := 0.5000000003\, a$$
$$sigma\_2 := -1.500000000\, a$$
$$sigma\_3 := -0.3333333333\, a$$

b) Obtener el valor de a para que la densidad de energía elástica debida a las componentes desviadoras valga Wd = 1E3 J/m3.

```
> sigma_m := 1/3*Trace(sigma);
```

$$sigma\_m := -0.4444444443\, a$$

```
> epsilon_v := Trace(epsilon);
```

$$epsilon\_v := -0.0004444444445$$

```
> sigma_d := sigma-sigma_m*IdentityMatrix(3);
```

$$sigma\_d := \begin{bmatrix} 0.4444444443\, a & \frac{a\sqrt{3}}{2} & 0 \\ \frac{a\sqrt{3}}{2} & -0.5555555557\, a & 0 \\ 0 & 0 & 0.1111111110\, a \end{bmatrix}$$

```
> epsilon_d := epsilon-1/3*epsilon_v*IdentityMatrix(3);
```

$$epsilon\_d := \begin{bmatrix} 0.000592592592551851845, & 0.00115470053800000006, & 0. \\ 0.00115470053800000006, & -0.000740740740748148190, & 0. \\ 0., & 0., & 0.000148148148151851835 \end{bmatrix}$$

```
> Wd :=0;
  for i from 1 to 3 do;
    for j from 1 to 3 do;
      Wd := Wd + 1/2*sigma_d[i,j]*epsilon_d[i,j];
    od;
  od;
  Wd;
```

$$Wd := 0$$
$$0.0003456790124\, a + 0.0005773502690\, a\sqrt{3}$$

```
> solve(Wd=1E3);
```

$$743119.2662$$

Problema 5

Ver enunciado entregado.

a) Calcular las componentes de la deformación y la deformación volumétrica en cualquier punto.

```
> restart;
> u := (x,y)->alpha[1]+alpha[2]*x+alpha[3]*y;
```

$$u := (x, y) \to \alpha_1 + \alpha_2\, x + \alpha_3\, y$$

```
> v := (x,y)->alpha[4]+alpha[5]*x+alpha[6]*y;
```

$$v := (x, y) \rightarrow \alpha_4 + \alpha_5\, x + \alpha_6\, y$$

Para obtener los parametros basta con tres puntos del cuadrilátero (A, B y D)

```
> ecu_dis := {u(0,0)=1, v(0,0)=0, u(50,0)=1.1, v(50,0)=0,
  u(0,20)=1.1, v(0,20)=-1};
```

$$ecu\_dis := \{\alpha_1 = 1, \alpha_4 = 0, \alpha_1 + 50\, \alpha_2 = 1.1, \alpha_4 + 50\, \alpha_5 = 0, \alpha_1 + 20\, \alpha_3 = 1.1, \alpha_4 + 20\, \alpha_6 = -1\}$$

```
> solu_dis :=
  solve(ecu_dis,{alpha[1],alpha[2],alpha[3],alpha[4],alpha[5
  ],alpha[6]});
```

$$solu\_dis := \{\alpha_5 = 0., \alpha_1 = 1., \alpha_4 = 0., \alpha_3 = 0.005000000000, \alpha_2 = 0.002000000000, \alpha_6 = -0.05000000000\}$$

Comprobación en el otro vértice del cuadrilátero (C)

```
> subs(solu_dis,[u(50,20),v(50,20)]);
```

$$[1.200000000, -1.000000000]$$

Funciones que dan el campo de desplazamientos

```
> ud := (x,y)->subs(solu_dis,u(x,y));
```

$$ud := (x, y) \rightarrow \mathrm{subs}(solu\_dis, \mathrm{u}(x, y))$$

```
> vd := (x,y)->subs(solu_dis,v(x,y));
```

$$vd := (x, y) \rightarrow \mathrm{subs}(solu\_dis, \mathrm{v}(x, y))$$

```
> with(LinearAlgebra):
```

Gradiente de desplazamientos

```
> grad_uv :=
  Matrix([[diff(ud(x,y),x),diff(ud(x,y),y)],[diff(vd(x,y),x)
  ,diff(vd(x,y),y)]]);
```

$$grad\_uv := \begin{bmatrix} 0.002000000000 & 0.005000000000 \\ 0 & -0.05000000000 \end{bmatrix}$$

Deformación lineal 2D

```
> epsilon2 := 1/2 * (grad_uv+Transpose(grad_uv));
```

$$\varepsilon 2 := \begin{bmatrix} 0.0020000000000000000004 & 0.0025000000000000000004 \\ 0.0025000000000000000004 & -0.0500000000000000000028 \end{bmatrix}$$

Deformación lineal 3D, teniendo en cuenta deformación plana

```
> epsilon := Matrix(3,3,epsilon2);
```

$$\varepsilon := \begin{bmatrix} 0.0020000000000000000004 & 0.0025000000000000000004 & 0. \\ 0.0025000000000000000004 & -0.0500000000000000000028 & 0. \\ 0. & 0. & 0. \end{bmatrix}$$

```
> e_v := Trace(epsilon);
```

$$e\_v := -0.04800000000$$

```
> e_des := epsilon-1/3*e_v;
```

$$e\_des := \begin{bmatrix} 0.0179999999999999988\,, & 0.0025000000000000000004\,, & 0. \\ 0.0025000000000000000004\,, & -0.0340000000000000000024\,, & 0. \\ 0.\,, & 0.\,, & 0.0160000000000000000002 \end{bmatrix}$$

b) Hallar las fuerzas P y H que transmite la máquina al apoyo, por metro de anchura (la anchura se mide en la dirección OZ).

```
> sigma := lambda*e_v*DiagonalMatrix(Vector(3,1))+2*mu*epsilon;
```

$$\sigma := \begin{bmatrix} -0.04800000000\,\lambda + 0.004000000000\,\mu, & 0.005000000000\,\mu, & 0. \\ 0.005000000000\,\mu, & -0.04800000000\,\lambda - 0.1000000000\,\mu, & 0. \\ 0., & 0., & -0.04800000000\,\lambda \end{bmatrix}$$

```
> datos_mat:={E=1000,nu=0.4};
```

$$datos\_mat := \{E = 1000, \nu = 0.4\}$$

```
> param := {lambda = subs(datos_mat,E*nu/((1+nu)*(1-2*nu))), mu = subs(datos_mat,E/(2*(1+nu)))};
```

$$param := \{\lambda = 1428.571428, \mu = 357.1428571\}$$

```
> sigma1 := map2(subs,param,sigma);
```

$$\sigma 1 := \begin{bmatrix} -67.14285711 & 1.785714286 & 0. \\ 1.785714286 & -104.2857142 & 0. \\ 0. & 0. & -68.57142854 \end{bmatrix}$$

```
> n1 := <0,1,0>;
```

$$n1 := \begin{bmatrix} 0 \\ 1 \\ 0 \end{bmatrix}$$

```
> t1 := sigma1.n1;
```

$$t1 := \begin{bmatrix} 1.78571428599999993 \\ -104.285714200000001 \\ 0. \end{bmatrix}$$

Reacciones en kg

```
> V := t1.n1*100*50;
```

$$V := -521428.5710$$

```
> H := t1[1]*100*50;
```

$$H := 8928.571430$$

c) Determinar la deformación normal en la dirección de la diagonal AC.

```
> rAC0:=<50,20>;
```

$$rAC0 := \begin{bmatrix} 50 \\ 20 \end{bmatrix}$$

```
> LAC0:=Norm(rAC0,Euclidean);
```

$$LAC0 := 10\sqrt{29}$$

```
> nAC0 := (1/LAC0)*rAC0;
```

$$nAC0 := \begin{bmatrix} \dfrac{5\sqrt{29}}{29} \\ \dfrac{2\sqrt{29}}{29} \end{bmatrix}$$

```
> epsilon_d:=nAC0.(epsilon2.nAC0);
```

$$epsilon\_d := -0.003448275861$$

e) Hallar la tensión máxima que se produce en el neopreno, indicando el plano en el que se produce.

```
> Eigenvectors(sigma1);
```

$$\begin{bmatrix} -67.0572029861242100+0.\,I \\ -104.371368323875785+0.\,I \\ -68.5714285399999994+0.\,I \end{bmatrix}, \begin{bmatrix} 0.998851597857721152+0.\,I & -0.0479112247503374124+0.\,I & 0.+0.\,I \\ 0.0479112247503374124+0.\,I & 0.998851597857721152+0.\,I & 0.+0.\,I \\ 0.+0.\,I & 0.+0.\,I & 1.+0.\,I \end{bmatrix}$$

Problema 3 (Aplicaciones)

Sea un tubo cilíndrico grueso de longitud L, de radio interior a y radio exterior b, con deformació impedida en sus dos extremos +/- L/2. En la superficie exterior r = b la deformación se halla totalmente impedida, mientras que en la superficie r = a se aplica una presión interior p. Para una sección suficientemente alejada de los extremos, se desea obtener:

1) Presión ejercida en la superficie exterior r = b para restringir completamente el desplazamientc en la misma.
2) Desplazamientos ur en la superficie interior r = a.
3) Distribución de tensiones en la pared del tubo.

1) Presión ejercida en la superficie exterior r = b para restringir completamente el desplazamiento en la misma.

```
> restart:
> ur(r):=r/(2*(mu+lambda))*(a^2*p[1]-b^2*p[2])/(b^2-a^2)+(a^2*b^2)/(2*mu)*(1/r)*(p[1]-p[2])/(b^2-a^2);
```

$$\mathrm{ur}(r) := \frac{r\,(a^2 p_1 - b^2 p_2)}{(2\,\mu + 2\,\lambda)\,(b^2 - a^2)} + \frac{1}{2}\,\frac{a^2\,b^2\,(p_1 - p_2)}{\mu\,r\,(b^2 - a^2)}$$

```
> ecu:=subs({r=b,p[1]=p},ur(r));
```

$$ecu := \frac{b\,(a^2 p - b^2 p_2)}{(2\,\mu + 2\,\lambda)\,(b^2 - a^2)} + \frac{1}{2}\,\frac{a^2\,b\,(p - p_2)}{\mu\,(b^2 - a^2)}$$

```
> sol1 := p[2]=solve(ecu,p[2]);
```

$$sol1 := p_2 = \frac{a^2\,p\,(2\,\mu + \lambda)}{b^2\,\mu + a^2\,\mu + a^2\,\lambda}$$

2) Desplazamientos ur en la superficie interior r = a.

```
> sol2 := simplify(subs({sol1,p[1]=p},ur(r)));
> subs(r=a,sol2);
```

$$sol2 := -\frac{p\,a^2\,(r^2 - b^2)}{2\,r\,(b^2\,\mu + a^2\,\mu + a^2\,\lambda)}$$

$$-\frac{p\,a\,(-b^2 + a^2)}{2\,(b^2\,\mu + a^2\,\mu + a^2\,\lambda)}$$

```
> f1:=subs({mu=E/(2*(1+nu)),lambda=E*nu/((1+nu)*(1-2*nu))},{a=1,b=2,p=1e3,E=1e6,nu=0.3},sol2);
plot(f1,r=1..2,thickness=2);
```

$$f1 := -\frac{0.0002000000000\,(r^2-4)}{r}$$



3) Distribución de tensiones en la pared del tubo.

```
> sigma11 := r ->
  (a^2*p[1]-b^2*p[2])/(b^2-a^2)-(a^2*b^2)/(r^2)*(p[1]-p[2])/
  (b^2-a^2);
  sigma22 := r ->
  (a^2*p[1]-b^2*p[2])/(b^2-a^2)+(a^2*b^2)/(r^2)*(p[1]-p[2])/
  (b^2-a^2);
```

$$\sigma 11 := r \to \frac{a^2\,p_1 - b^2\,p_2}{b^2-a^2} - \frac{a^2\,b^2\,(p_1-p_2)}{r^2\,(b^2-a^2)}$$

$$\sigma 22 := r \to \frac{a^2\,p_1 - b^2\,p_2}{b^2-a^2} + \frac{a^2\,b^2\,(p_1-p_2)}{r^2\,(b^2-a^2)}$$

```
> f11:=subs([a=1,b=2,p[1]=1000,p[2]=500],sigma11(r));
  f22:=subs([a=1,b=2,p[1]=1000,p[2]=500],sigma22(r));
  plot([f11,f22],r=1..2,thickness=2);
```

$$f11 := -\frac{1000}{3} - \frac{2000}{3\,r^2}$$

$$f22 := -\frac{1000}{3} + \frac{2000}{3\,r^2}$$




```
> sigma11a:=simplify(subs({sol1,p[1]=p},sigma11(r)));
  sigma22a:=simplify(subs({sol1,p[1]=p},sigma22(r)));
```

$$sigma11a := -\frac{(r^2\,\mu + r^2\,\lambda + b^2\,\mu)\,a^2\,p}{(b^2\,\mu + a^2\,\mu + a^2\,\lambda)\,r^2}$$

$$sigma22a := -\frac{(r^2\,\mu + r^2\,\lambda - b^2\,\mu)\,a^2\,p}{(b^2\,\mu + a^2\,\mu + a^2\,\lambda)\,r^2}$$

```
> f11:=subs({mu=E/(2*(1+nu)),lambda=E*nu/((1+nu)*(1-2*nu))},
  {a=1,b=2,p=1000,E=1e6,nu=0.3},sigma11a);
  f22:=subs({mu=E/(2*(1+nu)),lambda=E*nu/((1+nu)*(1-2*nu))},
  {a=1,b=2,p=1000,E=1e6,nu=0.3},sigma22a);
  plot([f11,f22],r=1..2,thickness=2);
```

$$f11 := -\frac{0.0004000000000\,(961538.4616\,r^2 + 0.1538461538\;10^7)}{r^2}$$

$$f22 := -\frac{0.0004000000000\,(961538.4616\,r^2 - 0.1538461538\;10^7)}{r^2}$$



Problema 1 (Aplicaciones)*

Ver enunciado entregado.

1) Comprobar que la función ϕ generadora de tensiones dada cumple efectivamente las ecuaciones de equilibrio en el medio continuo.

```
> restart;
> with(LinearAlgebra):
> phi := a*x^3 + b*x^2*y + c*x*y^2 + d*y^3;
```

$$\phi := a\,x^3 + b\,x^2\,y + c\,x\,y^2 + d\,y^3$$

```
> sigma[xx] := diff(phi, y, y) + V;
```

$$\sigma_{xx} := 2\,c\,x + 6\,d\,y + V$$

```
> sigma[yy] := diff(phi, x, x) + V;
```

$$\sigma_{yy} := 6\,a\,x + 2\,b\,y + V$$

```
> sigma[xy] := -diff(phi, x, y);
```

$$\sigma_{xy} := -2\,b\,x - 2\,c\,y$$

```
> V := rho_c*g*y;
  B := -<diff(V,x),diff(V,y)>;
```

$$V := rho\_c\,g\,y$$

$$B := \begin{bmatrix} 0 \\ -rho\_c\, g \end{bmatrix}$$

```
> diff(sigma[xx], x) + diff(sigma[xy], y) + B[1];
```

$$0$$

```
> diff(sigma[xy], x) + diff(sigma[yy], y) + B[2];
```

$$0$$

2) Determinar mediante las condiciones de contorno y las ecuaciones de equilibrio del contorno los coeficientes (a, b, c) y la distribución de tensiones en la presa.

equilibrio de fuerzas horizontales: la resultante de la acci´on hidrost´atica sobre el paramento vertical debe ser equilibrada por la resultante del cortante en la base de la presa:

```
> ec1 := 1/2*h*rho_w*g*h = int(subs(y=-h, sigma[xy]), x=0..l); # equilibrio de fuerzas horizontales
```

$$ec1 := \frac{h^2\, rho\_w\, g}{2} = -b\, l^2 + 2\, c\, h\, l$$

equilibrio de fuerzas verticales: el peso de la presa debe ser equilibrada por la resultante de la tensi´on vertical en la base

```
> ec2 := -1/2*h*l*rho_c*g = int(subs(y=-h, sigma[yy]), x=0..l); # equilibrio de fuerzas verticales
```

$$ec2 := -\frac{h\, l\, rho\_c\, g}{2} = 3\, a\, l^2 - 2\, b\, h\, l - h\, l\, rho\_c\, g$$

la componente de tensi´on sigma_xx en punto inferior izquierdo debe igualar a la presi´on hidrost´atica

```
> ec3 := -rho_w*g*h = subs(y=-h, x=0, sigma[xx]); # sigma_xx en punto inferior izquierdo
```

$$ec3 := -rho\_w\, g\, h = -6\, d\, h - rho\_c\, g\, h$$

Por ´ultimo, la cuarta ecuaci´on la obtenemos imponiendo la condici´on de tensiones normales nulas en el paramento inclinado que est´a libre de acciones. En primer lugar obtenemos los vectores normal y tangencial unitario:

```
> n := Normalize(<h,l>,Euclidean,conjugate=false);
  t := <n[2],-n[1]>;
```

$$n := \begin{bmatrix} \dfrac{h}{\sqrt{h^2 + l^2}} \\ \dfrac{l}{\sqrt{h^2 + l^2}} \end{bmatrix}$$

$$t := \begin{bmatrix} \dfrac{l}{\sqrt{h^2 + l^2}} \\ -\dfrac{h}{\sqrt{h^2 + l^2}} \end{bmatrix}$$

Formulamos ahora la matriz de tensiones como una funci´on de la coordenada s que define la posici´on en dicho paramento:

```
> ss := s -> subs(x=s*t[1],y=s*t[2],<<sigma[xx], sigma[xy]>|<sigma[xy], sigma[yy]>>);
```

$$ss := s \rightarrow \mathrm{subs}(x = s\, t_1, y = s\, t_2, \langle\langle \sigma_{xx} \mid \sigma_{xy}\rangle \mid \langle \sigma_{xy}, \sigma_{yy}\rangle\rangle)$$

Aplicamos dicha funci´on al punto medio (E) del paramento

```
> ssE := ss(sqrt(h^2+l^2)/2);
```

$$ssE := \begin{bmatrix} c\,l - 3\,d\,h - \frac{rho\_c\,g\,h}{2} & -b\,l + c\,h \\ -b\,l + c\,h & 3\,a\,l - b\,h - \frac{rho\_c\,g\,h}{2} \end{bmatrix}$$

Calculamos la tensi´on normal:

```
> ssEn := simplify(n . (ssE . n),symbolic);
```

$$ssEn := -\frac{-6\,h^2\,c\,l + 6\,h^3\,d + h^3\,rho\_c\,g + 6\,h\,b\,l^2 - 6\,l^3\,a + l^2\,h\,rho\_c\,g}{2\,(h^2 + l^2)}$$

La ecuaci´on 4 ya puede expresarse anulando dicha tensi´on normal:

```
> ec4 := ssEn=0;
```

$$ec4 := -\frac{-6\,h^2\,c\,l + 6\,h^3\,d + h^3\,rho\_c\,g + 6\,h\,b\,l^2 - 6\,l^3\,a + l^2\,h\,rho\_c\,g}{2\,(h^2 + l^2)} = 0$$

Mediante ls funci´on solve se resuelven las cuatro inc´ognitas de la funci´on de tensiones:

```
> solu := solve({ec1, ec2, ec3, ec4}, {a, b, c, d});
```

$$solu :=$$

$$\{b = -\frac{h^2\,rho\_w\,g}{2\,l^2},\ d = \frac{1}{6}\,rho\_w\,g - \frac{1}{6}\,rho\_c\,g,\ a = \frac{g\,h\,(-2\,h^2\,rho\_w + l^2\,rho\_c)}{6\,l^3},\ c = 0\}$$

Sustituimos estos valores para calcular las componentes del campo de tensiones

```
> 'sigma[xx]' = simplify(subs(solu,sigma[xx]));
```

$$\sigma_{xx} = y\,rho\_w\,g$$

```
> 'sigma[yy]' = simplify(subs(solu,sigma[yy]));
```

$$\sigma_{yy} = \frac{g\,(-2\,h^3\,x\,rho\_w + h\,x\,l^2\,rho\_c - h^2\,rho\_w\,y\,l + rho\_c\,y\,l^3)}{l^3}$$

```
> 'sigma[xy]' = simplify(subs(solu,sigma[xy]));
```

$$\sigma_{xy} = \frac{h^2\,rho\_w\,g\,x}{l^2}$$

3) Calcular las tensiones en los puntos del paramento inclinado, comprobando que se cumplen en el mismo las condiciones de contorno libre y obteniendo la tensión normal en dirección normal al paramento.

```
> ssn := simplify(subs(solu, n.(ss(s).n)),symbolic);
```

$$ssn := 0$$

```
> sst := s -> simplify(subs(solu, t.(ss(s).t)),symbolic);
```

$$sst := s \to \mathrm{simplify}(\mathrm{subs}(solu,\,t\,.\,((\mathrm{ss}(s))\,.\,t)),\,symbolic)$$

```
> data :=
  {rho_w=1000,rho_c=2500,g=9.81,l=50,h=50,E=30e9,nu=1/4}:
> long := subs(data,sqrt(h^2+l^2));
  lx := subs(data,l); ly := subs(data,h);
```

$$long := \sqrt{5000}$$

$$lx := 50$$

$$ly := 50$$

```
> plot(subs(data,sst(s))/1e6,s=0..long,labels=['distancia_m'
  ,'tension_MPa']);
```



valores num´ericos de los par´ametros (a,b,c,d) de la soluci´on:

```
> solun := subs(data,solu);
```

$$solun := \{b = -4905.000000, d = -2452.500000, a = 817.5000000, c = 0\}$$

Obtenemos ahora la distribuci´on de tensiones verticales en la base de la presa, funci´on de la coordenada x:

```
> f1 := subs(op(solun),data,subs(y=-h,sigma[yy]));
```

$$f1 := 4905.000000\, x - 735750.0000$$

```
> smax := subs(x=lx,f1);
```

$$smax := -490500.0000$$

```
> plot(f1/1e6,x=0..lx,labels=['distancia_m','tension_MPa']);
```



4) Calcular las deformaciones en los puntos D y E de la figura. Tomar para ello los valores numéricos E = 30 GPa, nu = 0.25.

```
> Const := 1/E*<<1-nu^2, -nu*(1+nu), 0>|<-nu*(1+nu), 1-nu^2, 0>|<0, 0, 1+nu>>;
```

$$Const := \begin{bmatrix} \frac{1-\nu^2}{E} & -\frac{\nu(1+\nu)}{E} & 0 \\ -\frac{\nu(1+\nu)}{E} & \frac{1-\nu^2}{E} & 0 \\ 0 & 0 & \frac{1+\nu}{E} \end{bmatrix}$$

```
> sigma[xx];
```

$$2\,c\,x + 6\,d\,y + rho\_c\,g\,y$$

```
> sigmav0 := <sigma[xx],sigma[yy],sigma[xy]>;
  sigmav := subs(solu,sigmav0);
```

$$sigmav0 := \begin{bmatrix} 2\,c\,x + 6\,d\,y + rho\_c\,g\,y \\ 6\,a\,x + 2\,b\,y + rho\_c\,g\,y \\ -2\,b\,x - 2\,c\,y \end{bmatrix}$$

$$sigmav := \begin{bmatrix} 6\left(\frac{1}{6}\,rho\_w\,g - \frac{1}{6}\,rho\_c\,g\right)y + rho\_c\,g\,y \\ \frac{g\,h\,(-2\,h^2\,rho\_w + l^2\,rho\_c)\,x}{l^3} - \frac{h^2\,rho\_w\,g\,y}{l^2} + rho\_c\,g\,y \\ \frac{h^2\,rho\_w\,g\,x}{l^2} \end{bmatrix}$$

```
> r_E := t*sqrt(l^2+h^2)/2;
  sE := subs(x=r_E[1],y=r_E[2],sigmav);
```

$$r\_E := \begin{bmatrix} \frac{l}{2} \\ -\frac{h}{2} \end{bmatrix}$$

$$sE := \begin{bmatrix} -3\left(\frac{1}{6}\,rho\_w\,g - \frac{1}{6}\,rho\_c\,g\right)h - \frac{rho\_c\,g\,h}{2} \\ \frac{g\,h\,(-2\,h^2\,rho\_w + l^2\,rho\_c)}{2\,l^2} + \frac{h^3\,rho\_w\,g}{2\,l^2} - \frac{rho\_c\,g\,h}{2} \\ \frac{h^2\,rho\_w\,g}{2\,l} \end{bmatrix}$$

```
> sEn := subs(op(solu),data,sE);
```

$$sEn := \begin{bmatrix} -245250.0000 \\ -245250.0000 \\ 245250.0000 \end{bmatrix}$$

```
> epsilon_E := Const . sEn;
  epsilon_En := subs(data,epsilon_E);
```

$$epsilon\_E := \begin{bmatrix} -\frac{245250.0000\,(1-\nu^2)}{E} + \frac{245250.0000\,\nu\,(1+\nu)}{E} \\ -\frac{245250.0000\,(1-\nu^2)}{E} + \frac{245250.0000\,\nu\,(1+\nu)}{E} \\ \frac{245250.0000\,(1+\nu)}{E} \end{bmatrix}$$

$$epsilon\_En := \begin{bmatrix} -0.5109375002\;10^{-5} \\ -0.5109375002\;10^{-5} \\ 0.00001021875000 \end{bmatrix}$$

```
> sD := subs(op(solu),data,x=lx/2, y=-ly, sigmav);
```

$$sD := \begin{bmatrix} -490500.0000 \\ -613125.0000 \\ 245250.0000 \end{bmatrix}$$

```
> epsilon_D := subs(data,Const.sD);
```

$$epsilon\_D := \begin{bmatrix} -0.8941406254\ 10^{-5} \\ -0.00001405078125 \\ 0.00001021875000 \end{bmatrix}$$

```
>
```